
Session 8 

Radio Frequency (RF) 
Radio antennas translate information into radio waves.  A radio wave is literally electric 

radiation all around us, going through buildings and even people, at varying speeds and power.  By 

sending information through radio waves, data is transmitted all over the world without wires.  Different 

frequencies can be used to transmit many different types of data at the same time in the same area. 

 
Most radio waves bounce around inside the atmosphere, though some escape into space.  What are the 

aliens listening to right now? 

The frequency (or speed) of the data, known as the Hertz (times per second) determines how 

many times the waves go up and down each second.  To send data, a transmitter sends pulses of 

electromagnetic radiation at a specific frequency.  The receiver is listening for data to come in at the 

assigned frequency, and if the right pulses of power are received, the receiving circuit can translate the 

pulses into bits of information. 

 
Wireless phones and internet routers work at 2.4 million Hertz, the same frequency as microwave ovens. 

Smart Car Version 4 - Wireless Control 
 Watching the robot run on its own is fun and all, but it’s time to take control for ourselves!  For 

that, we’ll need a remote control system.  To really ramp up our bots, we’re breaking into the 

unlicensed 2.4 GHz radio spectrum, giving us fast and long distance communications! 

nRF24L01+ 
This inexpensive little module is a computer within itself, and does all the hard work for you.  It 

even has a function to automatically check if the transmitted data is received at the other end! 

“The Nordic nRF24L01+ is a highly integrated, ultra low power (ULP) 2Mbps RF transceiver IC for 

the 2.4GHz ISM (Industrial, Scientific and Medical) band. With peak RX/TX currents lower than 14mA, a 

sub μA power down mode, advanced power management, and a 1.9 to 3.6V supply range, the 

nRF24L01+ provides a true ULP solution enabling months to years of battery life from coin cell or 

AA/AAA batteries. The Enhanced ShockBurst™ hardware protocol accelerator offloads time critical 

protocol functions from the application microcontroller enabling the implementation of advanced and 

robust wireless connectivity with low cost 3rd-party microcontrollers.” 

   - http://www.nordicsemi.com/eng/Products/2.4GHz-RF/nRF24L01P 

The low power version that we are using has a range of 50-200 feet (without interference).  

There are high-powered versions available with external antennas that can reach up to 2000 feet! 

 

n = Nordic Semiconductor Company 

RF = Radio Frequency 

24 = 2.4 GHz Band 

L = Low Power 

+ = New, Improved Version 

 

NRF Library 

Before we can use the nRF24L01+ with our Arduino, we need to include a “library” in our code 

(as we did with the Adafruit Motor Shield and the Sonar Sensor libraries).  In this case, there are several 

libraries to choose from; after some review, it appears the most popular and best suited to our needs 

comes from https://github.com/gcopeland/RF24. 

If you wish to complete this project at home, simply download the files from this site and place 

them in a folder called “RF24-master” within the “\Documents\Arduino\libraries” folder on your 

computer.  For additional information on installing libraries, see: http://arduino.cc/en/Guide/Libraries.  

http://www.nordicsemi.com/eng/Products/2.4GHz-RF/nRF24L01P
https://github.com/gcopeland/RF24
http://arduino.cc/en/Guide/Libraries

Addresses and Channels 
 

Notice the “pipe” commands in the programs: 

const uint64_t pipe = 0xE8E8F0F0E1LL; 

radio.openReadingPipe(1, pipe); 

radio.openWritingPipe(pipe); 

 

The nRF24 radios use "logical channels" for communications within a physical channel. Basically, 

a “pipe address” is sent ahead of every data transmission on a particular channel (frequency); a receiver 

of the same “pipe” listens for this address and upon detecting a match attempts to process the data 

packet which follows. (Source: https://github.com/natevw/node-nrf) 

Much like mail is sent to addresses on a street, the “pipe address” is used to communicate with 

similar devices over the air.  It consists of 10 characters, which are sandwiched between the symbols 

“0x” and “LL” in the code above; the “0x” indicates the value is in hexadecimal, and the “LL” means type 

“Long Long”, which is a large number up to 9,223,372,036,854,775,807 (the 10 hexadecimal character 

address we are using have 1,099,511,627,776 potential pipes to choose from). 

By using a 10-character code, we can assign a unique address value to each pair of transmitter 

and receiver and can use multiple systems in the same area. Each character in the code represents a 

number from 0 to 15 (A=10, B=11, C=12, D=13, E=14, F=15), which means theoretically we can have 

thousands of different remote systems working at the same time. In practice, interference is likely to get 

in the way with that many, but we should still be able to use several devices in the same area without 

too much trouble. 

 

 

 

 

Sample Good Pipe Addresses to Use: 

0x4664CD477ELL 0x8794B64D9CLL 0x43DD6D6BE4LL 0xB8E3CD6BDBLL 0x9BB98744EDLL 

0xED8E76774BLL 0x89397B6BECLL 0x3B39E8D93ELL 0xC98D94CC7CLL 0x989DDB8647LL 

0x87ED38984DLL 0xB6993D4ED8LL 0xB33E8E9CB3LL 0xDBB7C76CB9LL 0xD68B69C468LL 

0xB43D33B484LL 0xD679C64E96LL 0xC9E7448D77LL 0x837C4DCEEELL 0xEE686C9E93LL 

 

Note: According to the author of the NRF library code we are using, it is best practice to avoid certain 

patterns and characters, including 1, 2, 5, and A.  See http://maniacalbits.blogspot.com/ for more details 

https://github.com/natevw/node-nrf
http://maniacalbits.blogspot.com/

Transmitter Hardware 

Our controller hardware is pretty straight forward: plug the Joystick Shield onto the Arduino, 

then plug in the nRF24L01+ module onto the Joystick Shield.  We will only be using the joystick part of 

the controller board today, but in the future, we can use this board to do all sorts of awesome stuff! 

Receiver Hardware 

The wiring for the receiver on the car is a touch more complicated, though.  First, plug an 

nRF24L01+ module into the Base Module.  This base will help regulate power and is labeled to make it 

easier to connect the pins to the Adafruit Motor Shield.  With the modules connected, use Female-to-

Male jumper wires to connect the Base Module to the Adafruit Motor Shield as shown below.  

 

    

 

 

 

 

 

 

 

 

 

 

 

 

 

The wires connect like so: 

nRF24L01+ Arduino 

  VCC 5V . 

  GND GND . 

  CE 9 . 

  CSN 10 . 

  SCK 13 . 

  MO 11 . 

  MI 12 . 

  IRQ 2 .  

Note: 

The colors may vary; do not worry if you have different color 

wires than shown. 

Just be sure that the correct pins from the nRF24L01+ Base are 

connected to the correct pins on the Adafruit Motor Shield! 

Transmitter Program 

Take a moment to note the highlighted text in the above code.  This is the “pipe address” that 

will be used for this pair of devices.  If you wish to use more than one robot in the same room at the 

same time, make sure you use a unique address for your transmitter. 

(Note: the receiver must be using the same “pipe address” as the transmitter, so make sure you change 

the line in the code for this in both programs to ensure they match!) 

 

For a list of sample addresses to choose from, see page 89. 

 

The receiver code on the following pages is quite the doozy.  Pay close attention to the 

comments (after the // marks) to follow along. 

(Reminder: we are still using the Adafruit Motor Shield from the last couple sessions.  See previous 

projects for wiring assistance.) 

#include <SPI.h> 
#include <nRF24L01.h> 
#include <RF24.h> 
 
#define RF_CE 9    //Chip Enable pin, for transmit/receive 
#define RF_CSN 10  //Chip Select Not pin, used to determine listen/write state 
  
const uint64_t pipe = 0xE8E8F0F0E1LL; // Define the transmit pipe (LL=LongLong) 0xE8E8F0F0E1LL 
  
RF24 radio(RF_CE, RF_CSN); // Create a Radio object 
  
int command[4];  // 4 element array holding integers 
  
void setup() { 
  radio.begin();               // turn on radio 
  radio.openWritingPipe(pipe); // tell radio to use pipe address 
} 
 
void loop() { 
  command[1] = analogRead(A0); // read measurement of vertical Joystick 
  command[2] = analogRead(A1); // read measurement of horizontal Joystick 
 
  radio.write(command, sizeof(command));  // send data payload 
 
  delay(10); // wait 1/100th of a second before trying again 
} 

Receiver Program 

(continued on next page…) 

//RemoteControl Receiver v1.0 
//By Jason Groce 
//2015-04-22 
 
#include <SPI.h> 
#include <nRF24L01.h> 
#include <RF24.h> 
#include <Wire.h> 
#include <Adafruit_MotorShield.h> 
#include "utility/Adafruit_PWMServoDriver.h" 
Adafruit_MotorShield AFMS = Adafruit_MotorShield(); 
Adafruit_DCMotor *firstMotor = AFMS.getMotor(1);    //front right 
Adafruit_DCMotor *secondMotor = AFMS.getMotor(2);   //front left 
Adafruit_DCMotor *thirdMotor = AFMS.getMotor(3);    //back left 
Adafruit_DCMotor *fourthMotor = AFMS.getMotor(4);   //back right 
 
#define RF_CE 9     //Chip Enable pin, for transmit/receive 
#define RF_CSN 10   //Chip Select Not pin, used to determine listen/write state 
  
const uint64_t pipe = 0xE8E8F0F0E1LL;  // Define the transmit pipe (LL=LongLong) 0xE8E8F0F0E1LL 
  
RF24 radio(RF_CE, RF_CSN);   // Create a Radio object 
  
int command[4];        // 4 element array holding integers 
int JoystickX = 400;   // horizontal joystick measurement 
int JoystickY = 400;   // vertical joystick measurement 
int speedX = 0;   // used to calculate speed based on JoystickX 
int speedY = 0;   // used to calculate speed based on JoystickY 
int speedR = 0;   // calculated speed for right wheels 
int speedL = 0;   // calculated speed for left wheels 
int Lmin = 0;     // minimum measurement for lower joystick half 
int Lmax = 350;   // maximum measurement for lower joystick half 
int Hmin = 390;   // minimum measurement for upper joystick half 
int Hmax = 750;  // maximum measurement for upper joystick half 
 
void setup() { 
  radio.begin();                    // turn on radio 
  radio.openReadingPipe(1,pipe);   // tell radio to use pipe address 
  radio.startListening();;          // tell radio to start listening for messages 
  AFMS.begin();                     // turn on Motor Shield 
  firstMotor->setSpeed(speedY);    // front right motor 
  firstMotor->run(FORWARD); 
  secondMotor->setSpeed(speedY);  // front left motor 
  secondMotor->run(FORWARD); 

(continued on next page…) 

  thirdMotor->setSpeed(speedY);   // back left motor 
  thirdMotor->run(FORWARD); 
  fourthMotor->setSpeed(speedY);  // back right motor 
  fourthMotor->run(FORWARD); 
} 
  
void loop() { 
  if ( radio.available() ) {      // Read the data payload until we've received everything 
    bool done = false; 
    while (!done) { 
      done = radio.read(command,sizeof(command));    // Fetch the data payload 
    } 
    JoystickX = command[1];        // Copy the commands to the Joystick variables 
    JoystickY = command[2];  
  } 
 
  if( JoystickY > Hmin ) {           // if pointing up, move forward 
    speedY = map(JoystickY,Hmin,Hmax,0,255);   // map vertical joystick to motor speed 
    turn();                           // check to see if pointing left or right 
    firstMotor->run(FORWARD); 
    secondMotor->run(FORWARD); 
    thirdMotor->run(FORWARD); 
    fourthMotor->run(FORWARD); 
  } else if( JoystickY < Lmax ) {    // if pointing down, move backward 
    speedY = map(JoystickY,Lmax,Lmin,0,255);   // map vertical joystick to motor speed 
    turn();                           // check to see if pointing left or right 
    firstMotor->run(BACKWARD); 
    secondMotor->run(BACKWARD); 
    thirdMotor->run(BACKWARD); 
    fourthMotor->run(BACKWARD); 
  } else if ( JoystickX < Lmax ){    // if pointing due left, spin left 
    speedL = map(JoystickX,Lmax,Lmin,0,255);   // map horizontal joystick to motor speed 
    speedR = speedL; 
    firstMotor->run(FORWARD); 
    secondMotor->run(BACKWARD); 
    thirdMotor->run(BACKWARD); 
    fourthMotor->run(FORWARD); 
  } else if ( JoystickX > Hmin ){    // if pointing due right, spin right 
    speedR = map(JoystickX,Hmin,Hmax,0,255);   // map horizontal joystick to motor speed 
    speedL = speedR; 
    firstMotor->run(BACKWARD); 
    secondMotor->run(FORWARD); 
    thirdMotor->run(FORWARD); 
    fourthMotor->run(BACKWARD); 
  } else {                            // centered, stop 

    speedX = 0; 
    speedY = 0; 
    speedR = 0; 
    speedL = 0; 
  } 
   
  if ( speedL < 0 ) {  // check for incorrect reading and clear bad numbers 
    speedL = 0; 
  } 
  if ( speedR < 0 ) { 
    speedR = 0; 
  } 
   
  firstMotor->setSpeed(speedR); 
  secondMotor->setSpeed(speedL); 
  thirdMotor->setSpeed(speedL); 
  fourthMotor->setSpeed(speedR); 
   
  delay(10);    // wait 1/100th of a second to get next reading 
} 
void turn() { 
  if ( JoystickX < Lmax ) {                     // if joystick pointing left, go left 
    speedX = map(JoystickX,Lmax,Lmin,0,255);   // map horizontal joystick to motor speed value 
    speedL = speedY - speedX;                   // slow left wheels by amount joystick points left 
    speedR = speedY;                            // full speed right wheels 
  } else if ( JoystickX > Hmin ) {              // if joystick pointing right, go right 
    speedX = map(JoystickX,Hmin,Hmax,0,255);   // map horizontal joystick to motor speed value 
    speedR = speedY - speedX;                   // slow right wheels by amount joystick points right 
    speedL = speedY;                            // full speed left wheels 
  } else { 
    speedL = speedY;                            // full speed ahead! 
    speedR = speedY; 
  } 
} 

Sources 
https://arduino-info.wikispaces.com/Nrf24L01-2.4GHz-HowTo 

http://www.nordicsemi.com/eng/Products/2.4GHz-RF/nRF24L01P 

http://maniacalbits.blogspot.co.uk/2013/04/new-rf24-driver-release-fork.html 

http://maniacbug.github.io/RF24/FAQ.html 

http://www.diyembedded.com/tutorials/nrf24l01_0/nrf24l01_tutorial_0.pdf 

 

https://www.youtube.com/watch?v=1ibAiPNeQSI - Arduino IDE Basics and Installing the RF24 Library for 

the NRF24L01 Radio Module 

https://www.youtube.com/watch?v=UK1pNBqm6wc - Arduino Joystick Shield Nrf24l01 Wireless 2 dc 

motor control -- RC Car Project Part 1 

https://www.youtube.com/watch?v=dkgUP_tKqGo - Basic Arduino Based RC Controller for RC Vehicles 

 

http://rctractorguy.com/forum/topic/arduino-nrf24l01-radio-module-rc-control-tutorial-code 

https://github.com/nlinux1/nrf24l01--tx-rx-project1/blob/master/joystickTX.ino 

http://www.phoenixgarage.org/show_article/128 

http://arduino-for-beginners.blogspot.com/2013_02_01_archive.html 

http://maniacalbits.blogspot.com/2013/04/rf24-addressing-nrf24l01-radios-require.html 

 

https://arduino-info.wikispaces.com/Nrf24L01-2.4GHz-HowTo
http://www.nordicsemi.com/eng/Products/2.4GHz-RF/nRF24L01P
http://maniacalbits.blogspot.co.uk/2013/04/new-rf24-driver-release-fork.html
http://maniacbug.github.io/RF24/FAQ.html
http://www.diyembedded.com/tutorials/nrf24l01_0/nrf24l01_tutorial_0.pdf
https://www.youtube.com/watch?v=1ibAiPNeQSI
https://www.youtube.com/watch?v=UK1pNBqm6wc
https://www.youtube.com/watch?v=dkgUP_tKqGo
http://rctractorguy.com/forum/topic/arduino-nrf24l01-radio-module-rc-control-tutorial-code
https://github.com/nlinux1/nrf24l01--tx-rx-project1/blob/master/joystickTX.ino
http://www.phoenixgarage.org/show_article/128
http://arduino-for-beginners.blogspot.com/2013_02_01_archive.html
http://maniacalbits.blogspot.com/2013/04/rf24-addressing-nrf24l01-radios-require.html

